
Appendix A
Notation

We summarize here the notation that is used for describing the approximations of
the quasi-stationary distribution q and of the stationary distributions p(1) and p(0) of
the two auxiliary processes of the stochastic SIS model. All quantities included are
expressed in terms of the two essential parameters N and R0.

The first quantities to be defined have the subscript 1. They are important for
describing results when R0 is distinctly above one. The quantities μ̄1 and σ̄1 then
serve as mean and standard deviation, respectively, of the approximation of the
body and the near-end tails of the quasi-stationary distribution. Furthermore, the
expected time to extinction from quasi-stationarity then grows exponentially with
N. Its logarithm is asymptotically proportional to γ1N. The quantities γ1 and β1 are
related since β2

1 = 2γ1N. This implies that ϕ(β1) = exp(−γ1N)/
√

2π. Note also that
γ1 ≥ 0 for R0 > 0. We have

μ̄1 =
R0 −1

R0
N, (A.1)

σ̄1 =
√

N
R0

, (A.2)

y1(n) =
n− μ̄1

σ̄1
, (A.3)

z1(n) =
n− μ̄1

σ̄2
1

, (A.4)

β1 = sign(R0 −1)

√
2N

(
logR0 − R0 −1

R0

)
, (A.5)

γ1 = logR0 − R0 −1
R0

. (A.6)

184 A Notation

The next quantities have the subscript 2. They appear in approximations of the
left tail of the quasi-stationary distribution when R0 is distinctly above the value
one. They are

μ̄2 = N logR0, (A.7)

σ̄2 =
√

N, (A.8)

y2(n) =
n− μ̄2

σ̄2
, (A.9)

z2(n) =
n− μ̄2

σ̄2
2

, (A.10)

β2 =
√

N logR0. (A.11)

The study of the transition region requires a rescaling of R0 with the aid of a
quantity ρ defined as follows:

ρ = (R0 −1)
√

N. (A.12)

The transition region is characterized by the requirement that ρ = O(1) as N → ∞.
The next quantities have the subscript 3. They are useful in describing the quasi-

stationary distribution in the transition region:

μ̄3 = ρ
√

N, (A.13)

σ̄3 = σ̄2 =
√

N, (A.14)

y3(n) =
n− μ̄3

σ̄3
, (A.15)

z3(n) =
n− μ̄3

σ̄2
3

, (A.16)

β3 =
R0 −1

R0

√
N. (A.17)

It is useful to note that ρ = 0 when R0 = 1, and that ρ has the same sign as R0−1.
The quantities β1, β2, β3 have similar behaviors. They deviate slightly from each
other in the transition region, as shown by the following asymptotic approximations:

β1 ∼ ρ− 2
3

ρ2
√

N
, ρ = O(1), N → ∞, (A.18)

β2 ∼ ρ− 1
2

ρ2
√

N
, ρ = O(1), N → ∞, (A.19)

β3 ∼ ρ− ρ2
√

N
, ρ = O(1), N → ∞. (A.20)

The description of the uniform results in Chap. 12 uses the following notation:

μ̄12 =

{
μ̄1, R0 ≥ 1,

μ̄2, R0 ≤ 1,
(A.21)

σ̄12 =

{
σ̄1, R0 ≥ 1,

σ̄2, R0 ≤ 1,
(A.22)

y12(n) =

{
y1(n), R0 ≥ 1,

y2(n), R0 ≤ 1,
(A.23)

r0 =

{√
R0, R0 ≥ 1,

R0, R0 ≤ 1,
(A.24)

β0 = ρ, (A.25)

and

βi j =

{
βi, R0 ≥ 1, i = 0,1,2,3,

β j, R0 ≤ 1, j = 0,1,2,3.
(A.26)

We note that all of the quantities β0 = ρ, β1, β2, and β3 are asymptotic to ρ =
(R0 −1)

√
N in the transition region. Also, all of them are equal to zero when ρ = 0.

Thus, all of the quantities βi j are continuous functions of ρ. They are equal to zero
when ρ = 0, and they all have the same sign as ρ. This means that they are positive
for R0 > 1 and negative for R0 < 1.

The functions G1, G2, and G are defined as follows:

G1(z) =

⎧⎪⎪⎨
⎪⎪⎩

1− 1
z

+
1

exp(z)−1
, z �= 0,

1
2
, z = 0,

(A.27)

G2(x,z) =

⎧⎪⎨
⎪⎩

1− exp(−xz)
z

, z �= 0,

x, z = 0.

(A.28)

G(z) =

⎧⎪⎪⎨
⎪⎪⎩

1
z

log
exp(z)−1

z
, z �= 0,

1
2
, z = 0.

(A.29)

The functions H1, H∗
1 , H0, and H are defined as follows:

H1(y) =
Φ(y)
ϕ(y)

, (A.30)

H∗
1 (y) =

Φ∗(y)
ϕ∗(y)

, (A.31)

H0(y) =

⎧⎪⎨
⎪⎩

Ha(y), y ≤ ρb,

Ha(ρb)+
∫ y

ρb

H1(t)dt, ρ > ρb,
(A.32)

where the auxiliary function Ha is defined by

Ha(y) = − log |y|+
md

∑
k=1

(−1)k ak

2k
1

y2k , y < 0, (A.33)

and md is a positive integer depending on ρb. The Maple procedure in Appendix B
uses ρb = −3 and md = 4. The coefficients ak are defined by

ak =
(2k)!
k!2k , k = 0,1,2, . . . . (A.34)

Finally, the functions H, ρ̃, and R are defined as follows:

H(y) =
1

y + 1/H(y)

∫ y

−1/H(y)
H1(t)dt, (A.35)

ρ̃(y) = y +
1

H(y)
, (A.36)

and

R(N,y) = 1 +
ρ̃(y)√

N
= 1 +

y + 1/H(y)√
N

. (A.37)



Appendix B
A Maple Module for Numerical Evaluations

The numerical evaluations that are reported in this monograph have all been done
with Maple. We have used the Maple procedures that are included in the Maple
module named SIS given below. The module contains 21 procedures. We give some
brief comments on them.

We begin with six scalar functions. The normal distribution function Φ and the
normal density function ϕ are evaluated with the Maple functions PHI and phi,
respectively. Similarly, the three functions H, H0, and H1 are evaluated using the
Maple functions H, H0, and H1, respectively. In addition, the Maple procedure h is
used to determine H. It calls H1.

The remaining 15 procedures all deal with probability vectors. These probability
vectors are treated in the form of one-dimensional Maple Arrays. With the exception
of the last two procedures (ratio and short), the lengths of the vectors are N,
the population size for the SIS model. The first four of these procedures are PSI,
PSImn, rf, and cdf. The procedure PSI is used to determine the image under the
map Ψ of some vector p. A related procedure is the one called PSImn. It is used to
determine the image of the vector p(1) or p(0) after application of the map Ψ n times.
The ratio rk defined in (3.35) can be studied with the procedure rf. It is called by
the procedure PSI. Another useful procedure is called cdf. It converts a probability
vector p into its distribution function. It is called by both rf and by PSI.

There are three procedures associated with each of the two stationary dis-
tributions p(0) and p(1), and three more associated with the quasi-stationary
distribution q. First of all, each of these distributions can be evaluated numerically
with the aid of the procedures called p0f, p1f, and qsf, respectively. Furthermore,
we note that uniform approximations of body and of left tail of these distributions
are given in Chap. 13. Procedures that evaluate these approximations numerically
are contained in p0bodyappr, p0tailappr, p1bodyappr, p1tailappr, qsbodyappr,
and qstailappr, respectively.

In addition, the module contains two procedures that can be useful in certain
circumstances. They are called ratio and short. The first of these takes a vector of
length M with components p1, p2, . . . , pM, and produces a vector of length M − 1

188 B A Maple Module for Numerical Evaluations

whose component k equals the ratio pk+1/pk. The last procedure shortens a Maple
Array to its first M elements. It can be useful in plotting distributions.

We conclude this appendix by a listing of the Maple module named SIS:

SIS:=module()
export PHI,phi,H,H0,H1,h,PSI,PSImn,rf,cdf,
p0f,p1f,qsf,p0bodyappr,p0tailappr,p1bodyappr,
p1tailappr,qsbodyappr,qstailappr,ratio,short:

option package;
description "The qsd of the SIS model, and related
quantities";

PHI:=y->(1+erf(y/sqrt(2)))/2;

phi:=y->exp(-yˆ2/2)/sqrt(2*Pi):

H:=x->1/fsolve(h(x,v)=1,v):

H0:=proc(x)
local y;
if evalf(x)<= -3 then
evalf(-log(abs(x))-1/2/xˆ2+3/4/xˆ4-15/6/xˆ6

+105/8/xˆ8);
else
evalf(-log(3)-835/17496 + Int(H1(y),y=-3..x));

fi;
end proc:

H1:=x->PHI(x)/phi(x):

h:=proc(x,v)
local y;
if v=-x then
-x*H1(x);

else
(v/(x+v))*evalf(Int(H1(y),y=-v..x));

fi;
end proc:

PSI:=proc(N,R0,p)
local p0,pin,r,S,pp,ppsum:
description ‘‘Determine the image under the map PSI
of the vector p’’:

p0:=p0f(N,R0):

pin:=p0/p0[1]:
r:=rf(N,R0,p):
S:=cdf(r):
pp:=pin*S:
ppsum:=add(pp[k],k=1..N):
pp:=pp/ppsum:

end proc:

PSImn:=proc(N,R0,m,n)
local p0,p1,pin,p,k:
description ‘‘Determines PSIˆn(pˆ(m)), m=0 or 1’’:
p0:=p0f(N,R0):
p1:=p1f(N,R0):
pin:=p0/p0[1]:
if m=0 then
p:=p0:

else
p:=p1:

fi:
if n>0 then
for k from 1 to n do
p:=PSI(N,R0,p):

od:
fi:
p:

end proc:

rf:=proc(N,R0,p)
local p1,rho,pcum,rr,k:
description ‘‘The ratio r with vector p in
the numerator’’:

p1:=p1f(N,R0):
rho:=p1/p1[1]:
pcum:=cdf(p):
rr:=Array(1..N,1):
for k from 2 to N do
rr[k]:=(1-pcum[k-1])/rho[k]:

od:
rr:

end proc:

cdf:=proc(p)
local F,N,k;
description "The sum of p[j] from 1 to k":

N:=op(2,ArrayDims(p));
F:=Array(1..N);
F[1]:=p[1];
for k from 2 to N do
F[k]:=F[k-1]+p[k];

od;
F;

end proc;

p0f:=proc(N,R0)
local p0,k0,k,sump0;
description "The stationary distribution p0":
p0:=Array(1..N,1.);
k0:=max(1,floor((R0-1)*N/R0));
for k from k0+1 to N do
p0[k]:=p0[k-1]*(1-1/k)*(1-(k-1)/N)*R0;

od;
for k from k0-1 by -1 to 1 do
p0[k]:=p0[k+1]*(1+1/k)/R0/(1-k/N);

od;
sump0:=add(p0[k],k=1..N):
p0:=p0/sump0;
p0:

end proc:

p1f:=proc(N,R0)
local p1,k0,k,sump1;
description "The stationary distribution p1":
p1:=Array(1..N,1.):
k0:=max(1,floor((R0-1)*N/R0));
for k from k0+1 to N do
p1[k]:=p1[k-1]*(1-(k-1)/N)*R0;

od;
for k from k0-1 by -1 to 1 do
p1[k]:=p1[k+1]/R0/(1-k/N);

od;
sump1:=add(p1[k],k=1..N):
p1:=p1/sump1:

end proc:

qsf:=proc(N,R0)
local qs1,kvot,qs2,kvotmax,kvotmin,kv:
description "The quasi-stationary distribution":
if R0>1 then

qs1:=p0f(N,R0):
else
qs1:=p1f(N,R0):

fi:
kv:=2:
while kv > 1+1e-6 do
qs2:=PSI(N,R0,qs1):
kvot:=qs1/qs2:
kvotmax:=max(kvot):
kvotmin:=min(kvot):
kv:=max(kvotmax,1/kvotmin):
qs1:=qs2:
od:
qs2:
end proc:

p0bodyappr:=proc(N,R0)
local rho,beta2,beta3,mu1,sigma1,mu2,sigma2,a,
beta30,beta32,mu12,sigma12,r0,aa,f,p0:

description ‘‘Approximation of the body of p0’’:
rho:=(R0-1)*sqrt(N):
beta2:=sqrt(N)*log(R0):
beta3:=(R0-1)*sqrt(N)/R0:
mu1:=(R0-1)*N/R0: sigma1:=sqrt(N/R0):
mu2:=N*log(R0): sigma2:=sqrt(N):
if R0>1 then
r0:=1: beta30:=beta3: beta32:=beta3: mu12:=mu1:
sigma12:=sigma1:

else
r0:=R0: beta30:=rho: beta32:=beta2: mu12:=mu2:
sigma12:=sigma2:

fi:
a:=sqrt(R0/r0):
aa:=evalf(a/(0.5*log(N)+H0(beta30))/phi(beta32)):
f:=n->aa*evalf(phi((n-mu12)/sigma12)/n):
p0:=Array(1..N,f):

end proc:

p0tailappr:=proc(N,R0)
local rho,beta1,beta2,beta3,mu2,sigma2,a,beta10,
aa,f,p0:

description ‘‘Approximation of the left tail of
p0’’:

rho:=(R0-1)*sqrt(N):

beta1:=sqrt(2*N*(log(R0)-(R0-1)/R0)):
beta2:=sqrt(N)*log(R0):
beta3:=(R0-1)*sqrt(N)/R0:
mu2:=N*log(R0): sigma2:=sqrt(N):
if R0>1 then
a:=beta3/beta1: beta10:=beta1:

else
a:=1: beta10:=rho:

fi:
aa:=evalf(a/(0.5*log(N)+H0(beta10))/phi(beta2)):
f:=n->aa*evalf(phi((n-mu2)/sigma2)/n):
p0:=Array(1..N,f):

end proc:

p1bodyappr:=proc(N,R0)
local beta1,beta2,beta3,mu1,sigma1,mu2,sigma2,
beta12,beta13,mu12,sigma12,a,f,p1:

description ‘‘Approximation of the body of p1’’:
beta1:=sqrt(2*N*(log(R0)-(R0-1)/R0)):
beta2:=sqrt(N)*log(R0):
beta3:=(R0-1)*sqrt(N)/R0:
mu1:=(R0-1)*N/R0: sigma1:=sqrt(N/R0):
mu2:=N*log(R0): sigma2:=sqrt(N):
if R0>1 then
beta12:=beta1: beta13:=beta1: mu12:=mu1:
sigma12:=sigma1:

else
beta12:=beta2: beta13:=beta3: mu12:=mu2:
sigma12:=sigma2:

fi:
a:=evalf(1/H1(beta13)/sigma12/phi(beta12)):
f:=n->a*evalf(phi((n-mu12)/sigma12)):
p1:=Array(1..N,f):
end proc:

p1tailappr:=proc(N,R0)
local beta1,beta2,beta3,mu2,sigma2,beta13,a,f,p1:
description ‘‘Approximation of the left tail of
p1’’:

beta1:=sqrt(2*N*(log(R0)-(R0-1)/R0)):
beta2:=sqrt(N)*log(R0):
beta3:=(R0-1)*sqrt(N)/R0:
mu2:=N*log(R0): sigma2:=sqrt(N):
if R0>1 then

beta13:=beta1:
else
beta13:=beta3:

fi:
a:=evalf(1/H1(beta13)/sigma2/phi(beta2)):
f:=n->a*evalf(phi((n-mu2)/sigma2)):
p1:=Array(1..N,f):

end proc:

qsbodyappr:=proc(N,R0)
local rho,R,beta2,mu1,sigma1,mu2,sigma2,beta02,r0,
mu12,sigma12,a,aa,f,qs:

description ‘‘Approximation of the body of qs’’:
rho:=(R0-1)*sqrt(N):
R:=evalf(1+(rho+1/H(rho))/sqrt(N)):
beta2:=sqrt(N)*log(R0):
mu1:= (R0-1)*N/R0: sigma1:=sqrt(N/R0):
mu2:=N*log(R0): sigma2:=sqrt(N):
if R0>1 then
r0:=1: beta02:=rho: mu12:=mu1: sigma12:=sigma1:

else
r0:=R0: beta02:=beta2: mu12:=mu2: sigma12:=sigma2:

fi:
a:=1/sqrt(R0*r0):
aa:=evalf(a/(1+rho*H(rho))/phi(beta02)):
f:=n->aa*evalf(phi((n-mu12)/sigma12)*(1-1/Rˆn)/n):
qs:=Array(1..N,f):
end proc:

qstailappr:=proc(N,R0)
local rho,R,beta1,beta2,beta10,mu2,sigma2,
a,aa,f,qs:

description ‘‘Approximation of the left tail of
qs’’:

rho:=(R0-1)*sqrt(N):
R:=evalf(1+(rho+1/H(rho))/sqrt(N)):
beta1:=sqrt(2*N*(log(R0)-(R0-1)/R0)):
beta2:=sqrt(N)*log(R0):
mu2:=N*log(R0): sigma2:=sqrt(N):
if R0>1 then
a:=rho/R0/beta1: beta10:=beta1:

else
a:=1/R0: beta10:=rho:

fi:

aa:=evalf(a/(1+beta10*H(*beta10))/phi(beta2)):
f:=n->aa*evalf(phi((n-mu2)/sigma2)*(1-1/Rˆn)/n):
qs:=Array(1..N,f):

end proc:

ratio:=proc(p)
local rat,k,M;
description "The ratio p[k+1]/p[k]":
M:=op(2,ArrayDims(p));
rat:=Array(1..M-1):
for k from 1 to M-1 do
rat[k]:=p[k+1]/p[k];

od;
rat:

end proc;

short:=proc(p,M)
local sh,k:
description "Shortens the Array p to its first M
elements":

sh:=Array(1..M):
for k from 1 to M do
sh[k]:=p[k]:

od:
sh:

end proc:

end module:

50. I. Nåsell, A new look at the critical community size for childhood infections, Theor. Pop. Biol.
67 (2005) 203–216.

51. T.J. Newman and J-B Ferdy and C Quince, Extinction times and moment closure in the
stochastic logistic process, Theor. Pop. Biol. 65 (2004), 115–126.

52. R. M. Nisbet and W. S. C. Gurney, Modelling Fluctuating Populations, Wiley, New York, 1982.
53. R. H. Norden, On the distribution of the time to extinction in the stochastic logistic ppopulation

model, Adv. Appl. Prob. 14 (1982), 687–708.
54. F. W. J. Olver, Asymptotics and Special Functions, Academic Press, New York, San Fransisco,

London, 1974.
55. F. W. J. Olver, D. W. Lozier, R. F. Boisvert, and C. W. Clark, eds., NIST Handbook of

Mathematical Functions, http://dlmf.nist.gov, 2010.
56. I. Oppenheim, K. E. Shuler, and G. H. Weiss, Stochastic theory of nonlinear rate processes

with multiple stationary states, Physica 88A (1977), 191–214.
57. O. Ovaskainen, The quasistationary distribution of the stochastic logistic model, J. Appl. Prob.,

38 (2001), 898–907.
58. L. Pearl and L. J. Reed, On the rate of growth of the population of the United States since 1790

and its mathematical representation, Proc. Natl. Acad. of Sci. USA 6, 275–288, 1920.
59. P. K. Pollett, Reversibility, invariance, and μ-invariance, Adv. Appl. Prob. 20 (1988), 600–621.
60. P. K. Pollett, The generalized Kolmogorov criterion, Stoch. Proc Appl. 33 (1989), 29–44.
61. P. K. Pollett, Quasi-Stationary Distributions: A Bibliography,

http://www.maths.uq.edu.au/˜pkp/papers/qsds/qsds.pdf, 2010.
62. B. J. Prendiville, Discussion on symposium on stochastic processes, J. Roy. Stat. Soc. Ser. B

11, 273, 1949.
63. S. Ramanujan, Question 294, J. Indian Math. Soc. 3 (1911), 128.
64. E. Renshaw, Modelling Biological Populations in Space and Time, Cambridge University

Press, Cambridge, 1991.
65. L. M. Ricciardi, Stochastic population theory: birth and death processes, Mathematical

Ecology, an Introduction (Hallam & Levin, eds.) 17, 155–190, 1980, Berlin: Springer Verlag.
66. M. Schmitz, Quasi-stationarität in einem epidemiologischen Modell,

http://wwwmath.uni-muenster.de/statistik/alsmeyer
/Diplomarbeiten/Schmitz_Manuela.pdf, 2006.

67. E. Seneta, Quasi-stationary behavior in the random walk with continuous time, Austr. J. Statist.
8 (1966), 92–98.

68. A. Singh and J.P. Hespanha, A derivative matching approach to moment closure for the
stochastic logistic model, Bull. Math. Biol. 69 no6 (2007), 1909–1925.

69. D. Sirl, H. Zhang, and P. Pollett, Computable bounds for the decay parameter of a birth-death
process, J. Appl. Prob. 44 (2007), 476–491.

70. M. Takashima, Note on evolutionary processes, Bull. Math. Stat. 7, 18–24, 1957.
71. H. M. Taylor and S. Karlin, An Introduction to Stochastic Modeling, Academic Press, San

Diego, 1993.
72. A. Tsoularis and J. Wallace, Analysis of logistic growth models, Math. Biosci. 179 (2002),

21–55.
73. E. A. van Doorn, Quasi-stationary distributions and convergence to quasi-stationarity of birth-

death processes, Adv. Appl. Prob. 23 (1991), 683–700.



198 References

74. P. F. Verhulst, Notice sur la loi que la population suit dans son accroisement, Corr. Math. Phys.
X, 113–121, 1838.

